
	  function f_IMS_OtherResponse_200_MessageHeaderTX(MessageHeader p_MessageHeader_Request,
                                                   template (omit) ContentType p_ContentType := omit,
                                                   template (omit) charstring p_EmergencyTelURI := omit,
                                                   charstring p_ContactUri := px_IMS_CalleeContactUri,
                                                   boolean p_PreconditionsRequired := false,
                                                   IMS_PreconditionIndication_Type p_PreconditionIndication := NoPreconditionIndication,
                                                   IMS_CallType_Type p_CallType := NORMAL_CALL)
    runs on IMS_PTC return template (value) MessageHeader
  { /* @sic R5-130683: common handling for INVITE and UPDATE sic@ */
    /* p_EmergencyTelURI .. shall be set e.g. to tsc_IMS_Emergency_TelUri for condition A6 in A.3.1 */
    /* @sic R5s130756 additional changes: p_IsGIBA removed sic@ */
    /* @sic R5s150237: parameter p_AdditionalOptionTagsForRequire sic@ */
    /* @sic R5s150692 change 17: parameter p_CallType  sic@ */
    /* @sic R5s150851: no consistency check for p_CallType and p_EmergencyTelURI sic@ */
    /* @sic R5-160794: p_AdditionalOptionTagsForRequire replaced by p_PreconditionIndication sic@ */
    /* @sic R5-174544: SessionId removed sic@ */
    var template (value) MessageHeader v_MessageHeader_Response := f_IMS_BaseResponse_MessageHeaderTX(p_MessageHeader_Request);  /* @sic R5s190882: f_IMS_BaseResponse_MessageHeaderTX sic@ */
    var charstring v_Method := p_MessageHeader_Request.cSeq.method;
    var boolean v_A1A3 := false;
    var boolean v_A18 := pc_IMS_Video_FeatureTag;
    var boolean v_A19 := false;
    var template (value) RouteBody_List v_RouteBodyList;
    var template (omit) Require v_Require := omit;
    var template (omit) SemicolonParam_List v_ContactParams := omit;

    if ((p_PreconditionIndication == PreconditionRequired)p_PreconditionsRequired and ispresent(p_ContentType)) {    /* @sic R5-160794 sic@ NOTE: responses use Require header even for session modifications */
      v_Require := cs_Require(tsc_OptionTagList_precondition);
    }
    
    select (v_Method) {
      case ("INVITE") {
        v_A1A3 := true;
        v_A19 := true;
      }
      case ("UPDATE") {
        v_A1A3 := true;
      }
    }
    v_MessageHeader_Response.require         := v_Require;                                          /* @sic R5-160794 sic@ */
    v_MessageHeader_Response.contentType     := p_ContentType;

    if (v_A1A3) {
      v_RouteBodyList := f_IMS_RouteSet_MO_Call_TX(p_CallType);             /* @sic R5s150692 change 17 sic@ */
      v_MessageHeader_Response.recordRoute   := cs_RecordRoute(v_RouteBodyList);
      v_ContactParams := f_SemicolonParam_List_Add_TX(v_ContactParams, cs_GenericParam("audio"));                     /* @sic R5-165943 sic@ */
      if (v_A18 and v_A19) {
        v_ContactParams := f_SemicolonParam_List_Add_TX(v_ContactParams, cs_GenericParam("video"));                   /* @sic R5-191117 sic@ */
      }
      v_MessageHeader_Response.contact       := cs_Contact(f_SIP_BuildSipUri_TX(p_ContactUri), v_ContactParams);
    }
    if (ispresent(p_EmergencyTelURI)) {    // Condition A6 @sic R5s150851: in case of A6 p_EmergencyTelURI shall be handed over from the calling side; p_CallType is not a criteria sic@
      v_MessageHeader_Response.pAssertedID   := cs_PAssertedId(f_SIP_BuildSipUri_lr_TX(valueof(p_EmergencyTelURI)));
    }
    return v_MessageHeader_Response;
  }

	  function f_IMS_OtherResponse_200_MessageHeaderRX(template (value) MessageHeader p_MessageHeader_Request,
                                                   template ContentType p_ContentType := omit,
                                                   boolean p_PreconditionRequiredForSDP := true,
                                                   IMS_PreconditionIndication_Type p_PreconditionIndication := PreconditionRequired) runs on IMS_PTC return template (present) MessageHeader
  { /* NOTE: Message is sent by the UE => conditions A2, A4 and A5 are applicable;
     *       For A4 (GIBA) pAccessNetworkInfo is optional */
    /* @sic R5-130683 sic@ */
    /* @sic R5s130756 additional changes: p_IsGIBA removed sic@ */
    /* @sic R5-160790, R5-160903, R5-160793: p_PreconditionIndication sic@ */
    /* @sic R5-162930: p_PreconditionIndication := PreconditionRequired => per default there is a Require header if the message contains an SDP sic@ */
    /* @sic R5-174544: SessionId removed sic@ */
    var template (present) MessageHeader v_MessageHeader_Response := f_IMS_BaseResponse_MessageHeaderRX(p_MessageHeader_Request);     /* @sic R5s200000: f_IMS_BaseResponse_MessageHeaderRX sic@ */
    var template (present) PAccessNetworkInfo v_PAccessNetworkInfo := cr_PAccessNetworkInfo(f_IMS_PTC_GetRanType());  /* acc. to A.3.1 condition A5: access network technology and, if applicable, the cell ID */
    var boolean v_A2A4 := false; /* true => response sent by UE for INVITE/UPDATE */
    var boolean v_A8 := true;    /* true => any response but not CANCEL */
    var boolean v_A20 := false;  /* true => response sent by UE for INVITE */
    var boolean v_A18 := pc_IMS_Video_FeatureTag;
    var charstring v_Method := valueof(p_MessageHeader_Request.cSeq.method);
    var integer v_Port_us := f_IMS_PTC_GetPort_us();  /* @sic R5s130985 Change 2 sic@ */
    var template (present) SemicolonParam_List v_MandatoryContactParams := {};
    var template (present) SipUrl v_ContactSipUrl := f_Contact_SipUri_HostPortRX(?, v_Port_us);             /* host address needs to be checked after receiving (f_IMS_MessageHeader_CheckContactAddr)
                                                                                                               @sic R5s150721, R5-153760: cr_SipUri_HostPort replaced by f_Contact_SipUri_HostPortRX sic@ */
    select (v_Method) {
      case ("UPDATE") {
        v_A2A4 := true;
      }
      case ("INVITE") {
        v_A2A4 := true;
        v_A20 := true;
      }
      case ("CANCEL") {
        v_A8 := false;
      }
    }
    
    if ((p_PreconditionIndication == PreconditionRequired)p_PreconditionRequiredForSDP and f_ContentType_IsSDP(p_ContentType)) { /* @sic R5-160794 sic@
                                                                                                        NOTE 1: in general a SIP response carries an SDP answer (i.e. no offer) and therefore uses a Require but no Supported header
                                                                                                        NOTE 2: when the SDP is optional, ContentType is optional too (-> "ifpresent") and  f_ContentType_IsSDP returns false */
      v_MessageHeader_Response.require := cr_Require(tsc_OptionTagList_precondition);                /* @sic R5-160790, R5-160903, R5-160793 sic@ */
    }
    
    if (not f_IMS_PTC_SecurityScheme_IsGiba() and v_A8) {                       /* @sic R5-130683: pAccessNetworkInfo is optional for GIBA; R5s140092 change 12: CANCEL  sic@
                                                                                   @sic R5-142223: new condition A8: Any response sent by the UE, except for CANCEL requests sic@ */
      v_MessageHeader_Response.pAccessNetworkInfo := v_PAccessNetworkInfo;
    }
    else {
      v_MessageHeader_Response.pAccessNetworkInfo := v_PAccessNetworkInfo ifpresent;
    }
    if (v_A2A4) {
      if (f_IMS_PTC_AudioFeatureTagIsRequired() ) {                                                                // @sic R5-173021 sic@ @sic R5-177127 sic@
        v_MandatoryContactParams := f_SemicolonParam_List_Add_RX(v_MandatoryContactParams, cr_FeatureParam_Audio); /* @sic R5-165943, R5-170564 sic@
                                                                                                                    * !!!! A11 is for "Response sent by UE", but when the response is not for INVITE or UPDATE
                                                                                                                    * !!!! there is no Contact header at all !!!! */
      }
      if (v_A18 and v_A20) {                                                                                       /* @sic R5-191117 sic@ */
        v_MandatoryContactParams := f_SemicolonParam_List_Add_RX(v_MandatoryContactParams, cr_FeatureParam_Video);
      }
      v_MessageHeader_Response.contact := cr_Contact(v_ContactSipUrl,
                                                     v_MandatoryContactParams);    /* @sic R5-165943 sic@ */
    }
  
    if (v_A20) {  /* response for INVITE
                     @sic R5-135004 Change 3 sic@
                     PROSE ISSUE: prose is not clear about whether there shall be any RecordRoute header for responses to requests other than INVITE */
      v_MessageHeader_Response.recordRoute := p_MessageHeader_Request.recordRoute;     // @sic R5-135004: same as in the INVITE sent to the UE sic@
    }

    v_MessageHeader_Response.contentType  := p_ContentType;

    return v_MessageHeader_Response;
  }

	  function f_IMS_MTCallSetup_Receive200OK_SendACK(IMS_InviteRequestWithSdp_Type p_InviteRequestWithSdp,
                                                  template SDP_Message p_SDP_Message := omit,
                                                  boolean p_PreconditionRequired200OK := true,
                                                  IMS_PreconditionIndication_Type p_PreconditionIndication := PreconditionRequired) runs on IMS_PTC return IMS_ResponseWithSdp_Type
  { /* @sic R5s130756 additional changes: p_IsGIBA removed sic@ */
    /* @sic R5w140112: p_InviteRequestWithSdp instead of p_InviteRequest sic@ */
    /* @sic R5w140112: returns IMS_ResponseWithSdp_Type instead of Response sic@ */
    /* @sic R5-160903: p_PreconditionIndication sic@ */
    /* @sic R5-162930: p_PreconditionIndication := PreconditionRequired => per default there is a Require header if the message contains an SDP sic@ */
    var InternetProtocol_Type v_ProtocolUsedForInvite := p_InviteRequestWithSdp.RoutingInfo.Protocol;
    var IMS_DATA_RSP v_IMS_DATA_RSP;
    var Response v_Response;
    var MessageHeader v_MessageHeader;
    var MessageHeader v_MessageHeader_Invite;
    var template (present) MessageHeader v_MessageHeader_OtherResponse_200;
    var template (value) MessageHeader v_MessageHeader_Ack;
    var template ContentType v_ContentType := omit;
    var template MessageBody v_MessageBody := omit;
    var SipUrl v_ContactUrl;
    var template (omit) SDP_Message v_SDP_Message := omit;

    if (ispresent(p_SDP_Message)) {
      v_ContentType := cr_ContentTypeSDP;
      v_MessageBody := cr_MessageBody_SDP;
    }

    // Step 12. Receive 200 OK. The UE responds to INVITE with a 200 OK final response after the user answers the call.
    v_MessageHeader_Invite := p_InviteRequestWithSdp.Invite.msgHeader;
    v_MessageHeader_OtherResponse_200 := f_IMS_OtherResponse_200_MessageHeaderRX(v_MessageHeader_Invite, v_ContentType, p_PreconditionRequired200OK, p_PreconditionIndication);    /* @sic R5-16090: p_PreconditionIndication sic@ */
    v_IMS_DATA_RSP := f_IMS_ReceiveResponse(v_MessageHeader_Invite, v_ProtocolUsedForInvite, cr_Response(c_statusLine200, v_MessageHeader_OtherResponse_200, v_MessageBody));  /* @sic R5s140350 sic@ */
    v_Response := v_IMS_DATA_RSP.Response;
    v_MessageHeader := v_Response.msgHeader;
    f_IMS_MessageHeader_CheckContactAddr(v_MessageHeader);      /* check whether contact contains "SIP URI with IP address or FQDN" (see @sic R5s120727 sic@) */
    v_SDP_Message := f_IMS_SIP_SdpMessageBody_DecodeMatchAndCheckSDP_Ifpresent(v_Response.messageBody, p_SDP_Message);

    if (isvalue(f_IMS_PTC_ImsInfo_DialogGetRemoteTag())) {      /* normal case: there has been e.g. a 183 response already
                                                                   @sic R5s140017 MCC160 implementation: use remote tag sic@ */
      f_IMS_MessageHeader_CheckTo(v_MessageHeader, dialogToTag);
    } else {                                                    /* special case (e.g. 16.X test cases): 200 OK is the only response */
      f_IMS_Dialog_CompleteCreationForMTCall(v_MessageHeader);  /* @sic R5s140017 MCC160 implementation: use remote tag sic@ */
    }
    
    // Step 13. Send ACK. The SS acknowledges the receipt of 200 OK for INVITE.
    v_ContactUrl := f_IMS_PTC_ImsInfo_GetContactUrl(dialog);    /* @sic R5s150908 change 4: dialog sic@
                                                                   @sic R5s170883: in some cases contact URI may be set just by 20o OK sic@  */
    v_MessageHeader_Ack := f_IMS_AckRequest_MessageHeaderTX(p_InviteRequestWithSdp.Invite);
    IMS_Client.send(cas_IMS_Ack_Request(cs_IMS_RoutingInfo(v_ProtocolUsedForInvite), cs_ACK_Request(v_ContactUrl, v_MessageHeader_Ack)));

    return valueof(cs_IMS_ResponseWithSdp(v_Response, v_SDP_Message));
  }





